

Excel Design Guidelines

By Neale Blackwood

a4@iinet.net.au

www.a4accounting.com.au



A4 ACCOUNTING
EXCEL CONSULTING, TRAINING AND WEBINARS

Contents

Guidelines	3
Structure.....	3
Formula Guidelines.....	4
Functions.....	4
Shortcut to Follow a Link.....	4
Use a Color Convention.....	4
Avoid Magic Numbers	4
One Formula to Rule Them All	6
Range Names.....	8
Make it a DATE	9
It's Good To TEXT	9
Two Other Useful Date Functions	10
Handling Errors.....	11
Structure Guidelines	12
Instructions.....	12
Index Sheet.....	12
Input Sheets	13
Data Sheets	13
Hiding Rows and Columns.....	14
Formatting.....	14
Linking	15
Hyperlinking	15
Validation Guidelines.....	16
Tolerance.....	16
General Guidelines	17
Use the QAT	17
Avoid the Split Screen	17
Learn shortcuts.....	17
Keep It Simple	17
Set Your Print Areas	18
Save Versions	18
Testing.....	18
Macros.....	18

Guidelines

It's difficult to define one set of rules for Excel that can be applied across the board in all situations. Excel is used in so many different industries; by so many different professions that a one size fits all set of rules is too restricting.

Some of these suggestions have been included in lists of Excel Best Practices. These are practices that have worked and continue to work for me. I have clients who have spreadsheets that I developed 15 years ago that are still using them successfully today.

This session is more about guidelines than strict rules. These are practices that I've used and seen used during the past 20 years that

- Reduce possible errors
- Simplify formulas
- Identify errors or omissions
- Incorporate structure in the file
- Allow flexibility
- Avoid common errors and problems
- Speed up navigation
- Improve maintenance
- Make it easier to read and understand a spreadsheet

For some of the guidelines I'll provide alternatives for those exceptional circumstances when you need to do something that I advise against.

Structure

Structure is the goal for spreadsheets. The more structured your spreadsheets, the more likely they can be easily maintained and amended. In my experience it's the unstructured files that cause the most problems and have the most issues.

When a spreadsheet starts out it has zero structure. It's just a blank grid. It is up to you to apply the structure. I don't need to convince Accountants to aim for structure. A well-structured GL and chart of accounts is a pleasure to work with. A badly structured chart of accounts is a nightmare.

Spreadsheets often have a life of their own. They start out being used for one report and develop over time to be used for more and more purposes. In most cases the more structured the file, the easier it is to add things.

I will be looking at four topic areas

- Formulas
- Structure
- Validation
- General / Other

Many of the topics are expanded on in my other webinars. Some of those webinars are freely available as a recording (including a manual and example files) on my website.

The companion file for this session includes many of these techniques.

Formula Guidelines

Functions

I will be recommending some functions and explaining how to use them. Other functions will be used for demonstration purposes only.

The focus of this session is about design techniques. If a function isn't explained you can use Excel's help system to learn more about it.

Shortcut to Follow a Link

Whether you create a spreadsheet from scratch or inherit a file, this tip can help you understand where the values come from.

Select a cell that is linked to another cell or a range. It doesn't matter if the link is in the current sheet or another sheet.

Hold the Ctrl key down and press [

That's the left square bracket (on the right of the letter P on the keyboard).

This will take you to the cell/range. You can press F5 then press Enter to return to the cell where you followed the link from.

Use a Color Convention

Colours are useful to draw user's attention to certain areas. It is best practice to use a color convention for input cells so that users are aware of the cells they should be changing. You can use Styles (Home ribbon tab) to keep your formatting consistent throughout a file. You can also use colours for sheet tabs, similar tabs are coloured the same as shown in the example file.

Avoid Magic Numbers

Magic numbers are those numbers that are hard-keyed into formula. Now some magic numbers are OK because they are part of how the formula works. The SUBTOTAL function includes a number as its first argument to define the type of calculation it performs.

```
=SUBTOTAL(9,B2:B10)
```

The 9 at the start defines a SUM function, the most common use of a SUBTOTAL function.

One common magic number that should be avoided is the Column Number in VLOOKUP functions. I prefer to replace the Column Number with a link to an entry cell or a MATCH function.

	E	F	G	H	I	J	K	L	M	N
1		Code	Description	Colour	Price					
2		1234	Widget	Red	12.50			2	3	4
3		1235	Widget	Green	12.75		Code	Description	Colour	Price
4		1236	Gadget	Red	13.50		1236	=VLOOKUP(\$K4,\$F\$2:\$I\$5,2,0)	Red	13.5
5		1237	Gadget	Green	13.75					

PRACTICE

1. The formula in L4 is =VLOOKUP(\$K4,\$F\$2:\$I\$5,2,0)
2. This formula can't be copied because of the hard-keyed 2 (arrowed in above image)
3. As you copy the formula across you need to amend the 2 to 3 and 4 respectively. Rather than hard-key the column number, use another cell to hold the column number - see the range L2:N2.
4. The formula can be amended to
 =VLOOKUP(\$K4,\$F\$2:\$I\$5,L2,0)
5. This formula can be copied across.
6. You can even replace the hard keyed cell L2 with a formula that automatically updates the column number based on the heading used in L3.
 =MATCH(L3,\$F\$1:\$I\$1,0)
7. This formula in L2 can be copied to M2 and N2. This works because the range in the MATCH function coincides with the width of the VLOOKUP table. MATCH returns the position number of the item within a range. It is perfect for Column Numbers in VLOOKUPs. It is commonly used with the INDEX function.

It's OK to key in numbers that won't change. There's always going to be 24 hours in a day and 12 months in a year. If there is any likelihood that a number could change then put it in a cell, label it and make it an input cell. You may never change it, but you if you need to, it's an easy change because it is in one place.

Having numbers keyed into formula makes them difficult to change. You may not be able to use find and replace to amend as you may inadvertently change cell references.

The recent potential company tax rate change is a good example of a rate that appeared to be fixed but may now change in the future. The GST rate may also increase.

One Formula to Rule Them All

In a report range you should aim to create one formula that can be copied down and across. That is not always easy and in some cases you may need to modify the structure of the sheet to make it possible.

In the example below I will take you through creating one formula that can be copied to three other report ranges. To do that I will use Helper Cells.

Helper Cells

I call them Helper Cells. The Column Number cells from the previous VLOOKUP example are Helper Cells. These are cells that may contain entries or formula. They are used to make your final formula simpler and easier to create and copy.

Helper cells can also be used to centralise some calculations. If a particular calculation is to be used in multiple formulas it can be more efficient and effective to do the calculation once and then refer to it in the other calculations.

	A	B	C	D	E	F
1						
2				Sales by Region	Jul	Aug
3				North	1,481	1,416
4				South	1,208	1,318
5				East	1,868	1,808
6				West	1,865	1,701
7						
8				Total	6,422	6,243
9						
10				Working Days (WA)	23	22
11	Helper Cells			Weekdays	23	22
12	Fixed References	Relative				
13				North	Jul	Aug
14	Working Days (WA)	North		Sales per Workday		
15	Weekdays	North		Sales per Weekday		
16						

PRACTICE

1. We need to create a formula in cell E14 that we can copy down and across and to the reports below.
2. The entries in columns A and B are Helper Cells and will allow us to create the single formula. The formula in A14 is `=D$10`, A15 is `=D$11`, B14 is `=D13` and B15 is `=D13`.

Formula Tips

Ctrl + ' to copy the formula exactly from above - not a relative copy

F4 - adds \$ sign

Copy formula within the Formula Bar

Select a range before using the Fill Handle (bottom right corner - cross)

3. The formula for cell E14 is

=SUMIF(\$D\$3:\$D\$11,\$B14,E\$3:E\$11)/SUMIF(\$D\$3:\$D\$11,\$A14,E\$3:E\$11)

Logic Formula

Not everyone is aware you can use logic calculations by themselves in a cell to return TRUE or FALSE. Hence, you can do logic testing in labelled cells that make it more apparent what is being done on the spreadsheet. Calculations or IF functions can then refer directly to the logic cells.

TRUE and FALSE can be used in numeric calculations. In Excel, TRUE is equal to 1 and FALSE is equal to zero.

Thus multiplying by TRUE will leave a value unchanged and multiplying by FALSE will zero a value.

	A	B	C	D	E	F	G	H
1			Jul	Aug	Sep	Oct	Nov	Dec
2		Cash - accounts	\$1,500	-\$2,000	-\$1,400	\$2,000	\$2,500	-\$1,600
3		Is Positive						
4		Is Negative						
5								
6		Balance Sheet extract						
7		Cash						
8								
9		Overdraft						

PRACTICE

1. We want to display the cash value in the correct category in the Balance Sheet, based on its value.
2. We will use rows 3 and 4 to determine if the value is positive or negative. The formula for C3 is =C2>=0. The formula for the C4 is =C2<0. These can be copied across.
3. We can then use formulas in rows 7 and 9 to display the correct value in the correct row. The formula for C7 is =C2*C3. The formula for C9 is =ABS(C2)*C4.

The ABS function converts all numbers to positives.

4. These can be copied across.
5. An alternative is to use the IF function.

C7 would be =IF(C3,C2,0)

C9 would be `=IF(C4,ABS(C2),0)`

Note we didn't use `=IF(C3=TRUE,C2,0)`

If the cell is already a logic result you can just use the cell itself.

Range Names

I use and highly recommend range names. Range names can simplify formula. They can reduce the need to use \$ signs to fix row and column references. In many cases they create self-documenting formula. Some users do not share my enthusiasm for range names and think they add another level of complexity to spreadsheets.


I have a two webinars (paid) dedicated to range names.

A quick example will demonstrate their power.

	A	B	C	D	E	F	G
1							
2			9%	9.25%	10%	9.75%	10%
3		\$45,000					
4		\$50,000					
5		\$55,000					
6		\$60,000					
7		\$65,000					

PRACTICE

- The normal formula for the above table in cell C3 is `=B3*C$2`.
- To use range names we can create two name ranges.
 B3:B7 can be named Income
 C2:G2 can be named Super
- To create a range name first select the range and then click in the Name Box to the left of the Formula Bar - type the name and press Enter.
- After creating the two names the formula in C3 can be `=Income*Super`
 This formula can be copied down and across and is identical in all cells.



Income		fx	45000
	A	B	
1			
2			
3		\$45,000	
4		\$50,000	
5		\$55,000	
6		\$60,000	
7		\$65,000	
8			

Make it a DATE

I recommend you add the DATE function to your list of useful functions. If you are using dates in your spreadsheet you can automate most of them using the DATE function.

Syntax

DATE(Year,Month,Day)

Year - a whole number representing the year

Month - a whole number representing the month. If the number is more than 12 it is treated as a sequential month. 12 is December, 13 is treated as January and it will increment the year as well.

Day - a whole number representing the day. If the number is more than the last day of the month it is treated as a day in the next month. A day value of 32 in January is equal to 1 and it will increment the month to February.

The DATE function is perfect for automatically increment dates by a year or a month. Incrementing by days can be done using normal formula.

As a general rule always use the first of the month for month dates.

Useful DATE formulas - assume a date in B4

Start of next month

=DATE(YEAR(B4),MONTH(B4)+1,1)

End of this Month

=DATE(YEAR(B4),MONTH(B4)+1,0)

Next Year same day

=DATE(YEAR(B4)+1,MONTH(B4),DAY(B4))

It's Good To TEXT

Excel handles text quite well. You can create automated headings. Unfortunately dates are one of the most common things you want to add to headings but there is trick to using dates in headings.

	A	B	C	D	E	F
1	Test	Report For		Heading (no TEXT)		Heading (with TEXT)
2	Month	1/08/2013		Report For 41487		Report For August 2013

To format dates and numbers to use with text entries you need to use the TEXT function.

The formula in cell D2 is =B1&" "&B2

The formula in cell F2 is =B1&" "&TEXT(B2,"mmmm yyyy")


The & symbol joins text together.

The formats you can use are the same as the ones in the Custom category of the Number Format →

They must be enclosed in inverted commas.

Two Other Useful Date Functions

As well as the DATE function there are two other functions worth adding to your skillset.



EOMONTH Function

This function can find the end of month date for any date.

Syntax

EOMONTH(Date,Num_Months)

Date is usually a cell reference to a date.

Num_Months is the number of months to add to the Date. Use 0 for the current month.

You can hack this function to calculate the first of the month as well.

	A	B	C
1	Dates	EOMONTH - current month	EOMONTH hack - start of month
2	23/06/2017	30/06/2017	1/06/2017
3	14/08/2017	31/08/2017	1/08/2017
4	2/05/2016	31/05/2016	1/05/2016

The formula in B2 is

=EOMONTH(A2,0)

The formula in C2 is

=EOMONTH(A2,-1)+1

EDATE Function

The EDATE function allows you add months to a date. This is useful for leases and loan agreements where you want to calculate the end date in four or five years' time.

Syntax

EDATE(Date,Num_Months)

Date is usually a cell reference to a date.

Num_Months is the number of months to add to the Date. Use 0 for the current month.

The EDATE function s handles February correctly - see rows 5 and 6 below.

	A	B	C
1	Dates	EDATE - one month	EDATE - five years
2	23/06/2017	23/07/2017	23/06/2022
3	14/08/2017	14/09/2017	14/08/2022
4	2/05/2016	2/06/2016	2/05/2021
5	29/01/2017	28/02/2017	29/01/2022
6	31/01/2017	28/02/2017	31/01/2022

The formula in B2 is

=EDATE(A2,1)

The formula in C2 is

=EDATE(A2,60)

Note: I have a dedicated free webinar on date calculations available on my website.

Handling Errors

Handling error in your model is important. The **IFERROR** function simplifies that process.it handles all errors the same. The function was added in Excel 2007.

Syntax

IFERROR(Reference, Value_if_error)

Reference – is typically a cell or a function/formula that may contain an error.

Value_if_error - is what to display or calculate if an error is encountered. This could be text – must be enclosed in quotation marks – a formula, or a value.

The IFERROR sheet tab has some examples.

	A	B	C
1	Formulas	IFERROR	Column B Formula
2	#DIV/0!		=IFERROR(A2,"")
3	#DIV/0!	0	=IFERROR(A3,0)
4	#N/A	missing	=IFERROR(A4,"missing")
5	#NAME?	Error	=IFERROR(A5,"Error")
6	#REF!		=IFERROR(A6,"")

Structure Guidelines

Getting the right structure in your sheet tabs can make development and maintenance easier.

In most cases its best to separate your processes into separate sheets. The types of sheets you would create include

- Instructions
- Index (navigation)
- Inputs
- Assumptions/Settings
- Data
- Reports
- Validation
- Workings
- Tables

Trying to do too much in a single sheet is a recipe for disaster. Excel's fixed row and column structure make it difficult to have multiple reports in the one sheet. Ideally you would separate your reports into different sheets so that you can get the rows and columns perfect for each report.

Colour code your sheet tabs. Keep similar sheets together.

Instructions

You should include instructions sheets for most of your files. This includes operational as well as maintenance instructions. I recommend including hyperlinks for those instructions that need users to go to another sheet. Consider using Textboxes for instructions with input sheets.

The shortcut to return from following a hyperlink is F5 + Enter – this is a handy tip to put on your Instructions sheet if you do use hyperlinks. Ctrl + k is the Hyperlink shortcut.

Cell Comments

Use cell comments to annotate cells and provide more information on what is required or where the information comes from. Cell comments are like sticky notes for cells and don't normally display.

Shift + F2 is the keyboard shortcut to add or edit a cell comment. You can also right click and Add or Edit a comment.

The Review Ribbon tab has an icon that shows all the comments in one step. You can also set a comment to display all the time.

Index Sheet

I recommend using specialised sheets for certain processes. This can lead to a lot of sheets. An Index sheet can make it easier to navigate around the file. I recommend having it as the very first sheet in the file which makes it easy to get to using the sheet navigation buttons.

You can even include a description of the sheet in the Index as a documentation feature.

Consider having a hyperlink back to the Index sheet in all the other sheets.

Input Sheets

Keep all your inputs in separate sheets to the reports. Or at least try to.

I have found this to be one of the hardest things to implement over the years. One of the problems is that you need to see the result of some inputs. If the inputs are in one sheet and the results are in another sheet than it can be hard to see the impact of your inputs.

One way around that is to display key results on the input sheet to enable users can immediately see the impact of their changes to the outputs.

Some of the inputs may not be expected to change and these may appear on a separate sheet dedicated to Settings.

Group your inputs either by frequency to change or the sequence they needed to entered.

If possible set up your inputs into a data layout. It's not pretty but it simplifies formula.

Compare the two layouts.

	A	B	C	
1	KPI	Jul-13	Aug-13	S
2	Credit Notes	15	10	
3	Customer Complaints	9	11	
4	New Customers	6	5	
5	New Products	1	2	
6				

	A	B	C
1	Month	KPI	Score
2	1/07/2013	Credit Notes	15
3	1/07/2013	Customer Complaints	9
4	1/07/2013	New Customers	6
5	1/07/2013	New Products	1
6	1/08/2013	Credit Notes	10
7	1/08/2013	Customer Complaints	11
8	1/08/2013	New Customers	5
9	1/08/2013	New Products	2

The data base layout on the right can be added to indefinitely. The table layout on the left is OK for a single year, but not for multiple years.

Use Data Validations to limit the entries in cells →

Shortcut to open the Data Validation dialog Alt + V + V


Data Sheets

Use separate sheets for separate data.

Within your data sheets avoid having blank rows or blank columns. These affect how easily Excel can automatically define tables.

Use the Format As Tables option as this has many more advantages than just formatting.

I have a free recorded webinar on data handling with manual and example files on my website.



Hiding Rows and Columns

If you need to hide rows and/or columns use Grouping to do it rather than just hiding them.


Grouping adds icons around the sheet and makes it obvious something is hidden. It also makes it quick to hide or unhide.

You can also set up hierarchies within your sheets to allow you to show/hide increasing amounts of detail. You can in effect have your summary and detail figures in the same sheet, in the same place.

You can add the Grouping icon to your QAT or use the following keyboard shortcuts.

Shift + Alt + right arrow - applies grouping

Shift + Alt + left arrow - removes grouping




If your report has SUM or SUBTOTAL functions in it, Excel can automatically create the Grouping using AutoOutline - in the Data Ribbon tab.

Formatting

Use formatting sparingly. Create a standard and be consistent in applying it.

Use a separate colour for input cells. Use Styles to make it easier to standardise formats different types of cells.

Turn off gridlines - looks cleaner, less visual noise. Apply our own borders.



Use the Format Painter - double click it to keep applying a format.

Linking

Linking internally within a file is necessary when you separate processes into separate sheets.

Linking to external files should be avoided. The reason is that if the files get separated in your system it will cause problems. The other aspect is that links can easily become corrupted if structural changes are made to the source file when the destination file is closed.

If you must do it then there are few ways to reduce the issues.

- Have a dedicated “link out” sheet in the source file. Bring all the entries that are used externally into the one sheet and use that sheet to link to.
- In the same way have a dedicated “link in” sheet in the destination sheet to centralise all the links coming in.
- Always have both files open when making structural changes and save both of them.
- Avoid using external references in formula - it makes it hard to break the links to the external file. Instead refer to a cell that has a link to the external file.
- Be very careful moving or copying charts between sheets - they are a common reason for “phantom” links between files.
- Be careful copying sheets between files they can bring in range names. Range names are another common source of “phantom” links.

Hyperlinking

When you use multiple sheets creating separate hyperlinks can be time consuming. You can use a formula to speed up the process.

See the INDEX sheet in the companion file - rows 13 and 14. Blog article explanation below.

<https://a4accounting.com.au/flexible-hyperlinks-in-excel/>

Validation Guidelines

Excel spreadsheets have a reputation for containing errors. Your job is to reduce those errors.

Including a validation sheet is probably one of the most important aspects of a spreadsheet. You can have validation cells throughout the spreadsheet but it is worthwhile to bring them all together into a central validation sheet.


This makes it easier to monitor all the validations because they are in one location.

I recommend using logic formula as described earlier to handle most of your validations.

I use TRUE to confirm a validation and FALSE to show something is not validating.

Use conditional formatting to colour FALSE cells red. See image below for settings.

The COUNTIF function is useful in dealing with large numbers of validations.



The formula in cell B4 is
=COUNTIF(B6:B20,FALSE)=0

See the Validation
sheet for examples.

Tolerance

When dealing in rounded numbers you will need to have tolerance levels. The ABS function is useful when working with tolerance amounts.

The ABS function only returns positives values. Hence you can use it to compare a difference amount to a positive tolerance value to see if it is within acceptable limits.

The Report sheet has example of using the ABS function.

Parameters

As well as validation checks which are black and white, you can also set up some reasonableness checks. These may be historic highs and lows for certain numbers or targets. If these are exceeded or not met it may be an indication that further investigation is warranted.

This could identify an error or a large transaction that happened on the last day of the month that affected certain ratios.

General Guidelines

These techniques didn't fit in to the other categories, but I have found them useful over the years.

Use the QAT

The QAT is the Quick Access Toolbar. Unfortunately many people are unaware of its existence. It hides above the Ribbon and people pretty much ignore it. This little toolbar can hold all your shortcuts and speed up your work and save you from hunting around in the Ribbon for that elusive icon.

I always move it below my Ribbon to make it easier to reach.

Some systems clear the QAT each time you log-on. What you can do is create your own personal QAT and then export it to save the settings. Then you can import it each morning to re-set it. If you make a change to the QAT remember to export it again to save those settings.

Right click any icon on the Ribbon to add it to the QAT. Right Click the QAT to customise it further.

You can also add macros to the QAT.

Avoid the Split Screen

Use Freeze Panes or separate Windows. You can open a new Window in the View Ribbon tab.

The shortcuts keys for Freeze Panes are

Excel 2007 onwards

Press in sequence (don't hold down) Alt + w f f

Excel 2003

Press in sequence (don't hold down) Alt + w f

Learn shortcuts

There are lots of ways to do things quicker in Excel and the less time you spend creating the more time you have to analyse.

I use the right click, right drag and Ctrl key techniques frequently.

I have free mouse and keyboard shortcut webinars available on my website.

Keep It Simple

This applies to formatting, charts and formula. Of course complexity can be relative. How complex a formula looks can depend on your Excel skill level.

Using Helper Cells can help simplify formula. Sometimes formulas have to be complicated, but at least try and make them as simple, or as short, as possible.

Set Your Print Areas

As well as setting print areas on any sheet that is likely to be printed make sure you check the Print Preview before printing to make sure you not going to take out a few forests to print your spreadsheet.

Excel 2010 changed the Print Preview dialog and some people don't like to the new one. You can use the old one. There is an icon you add to your QAT to open the old-fashioned Print Preview. It's called Print Preview Full Screen.

Save Versions

The F12 key is Save As and I recommend making regular versions. I do one per day when developing files and also one version per major revision that has been tested.

Testing

Make sure you test for things like empty cells, zeroes and error cells. IF functions are a common source of errors in Excel, make sure you test them.

Macros

If you really want to make the most of Excel I highly recommend you learn macros. Some things in Excel are only possible or practical via a macro. Visual Basic for Applications (VBA) is the macro language. It's a subset of Visual Basic programming language.

Some tasks in Excel can be extremely repetitive and time-consuming. Repetition can lead to errors. Macros are brilliant at repetitive tasks. There are few ways to create macros

- Record your own macros - recorded macros have many limitations - this is an easy process though and there are few tips to get the best results. No programming required.
- You can record and then edit your macros - this removes all of the limitations of recorded macros - this requires some programming skills
- Write your own macros from scratch - requires programming skills
- Copy macros off the web - there are lots of sources. Can be time consuming to find exactly the right one. You may need to modify it, so some programming skills are required.

If you are starting out from scratch I recommend Excel VBA For Dummies by John Walkenbach. There are different editions for different Excel versions. In my opinion he is the best Excel author.

I also have a series of paid webinars on macros if you are interested.